/**
 * @file HeaderMenu.qml
 * @brief Header menu view.
 * Copyright (c) 2013, Ford Motor Company
 * All rights reserved.
 *
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions are met:
 *
 * Redistributions of source code must retain the above copyright notice, this
 * list of conditions and the following disclaimer.
 *
 * Redistributions in binary form must reproduce the above copyright notice,
 * this list of conditions and the following
 * disclaimer in the documentation and/or other materials provided with the
 * distribution.
 *
 * Neither the name of the Ford Motor Company nor the names of its contributors
 * may be used to endorse or promote products derived from this software
 * without specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS"
 * AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
 * IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE
 * ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE
 * LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR
 * CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF
 * SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS
 * INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN
 * CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE)
 * ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE
 * POSSIBILITY OF SUCH DAMAGE.
 */

import QtQuick 2.0
import "../models/Constants.js" as Constants

Item {
    id: headerMenu
    anchors.fill: parent
    Text {        
        anchors.leftMargin: Constants.margin
        anchors.topMargin: Constants.margin
        anchors.left: parent.left
        anchors.verticalCenter: parent.verticalCenter
        color: Constants.primaryColor
        font.pixelSize: Constants.fontSize
        text: "75°"
    }

    Item {
        width: menuText.width
        height: parent.height
        anchors.topMargin: Constants.margin
        anchors.horizontalCenter: headerMenu.horizontalCenter
        signal clicked ()

        ClickableText {
            id: menuLogo
            anchors.horizontalCenter: menuText.horizontalCenter
            anchors.bottom: parent.verticalCenter
            text: "≡"
            defaultColor: Constants.primaryColor
            pressedColor: Constants.primaryColorPressed
            font.pixelSize: Constants.fontSize

            onClicked: parent.clicked()
            onPressed: menuText.press()
            onReleased: menuText.release()
            onCanceled: menuText.release()
        }

        ClickableText {
            id: menuText
            anchors.top: parent.verticalCenter
            text: "MENU";
            defaultColor: Constants.primaryColor
            pressedColor: Constants.primaryColorPressed
            font.pixelSize: Constants.fontSize

            onClicked: parent.clicked()
            onPressed: menuLogo.press()
            onReleased: menuLogo.release()
            onCanceled: menuLogo.release()
        }

        onClicked: {
            contentLoader.go("./views/MainMenuView.qml")
        }
    }

    Row {
        id: clock
        height: childrenRect.height
        anchors.right: parent.right
        anchors.verticalCenter: parent.verticalCenter
        anchors.rightMargin: Constants.margin
        anchors.topMargin: Constants.margin
        property date date: new Date()

        Timer {
            running: true
            repeat: true
            interval: 1000
            onTriggered: {
                clock.date = new Date();
            }
        }

        Text {
            text: Qt.formatTime(clock.date, "hh");
            color: Constants.primaryColor
            font.pixelSize: Constants.fontSize;
        }

        Item {
            width: 10
            height: parent.height
            Text {
                anchors.horizontalCenter: parent.horizontalCenter
                text: clock.date.getSeconds() % 2 ? ":" : ""
                color: Constants.primaryColor
                font.pixelSize: Constants.fontSize;
            }
        }

        Text {
            text: Qt.formatTime(clock.date, "mm");
            color: Constants.primaryColor
            font.pixelSize: Constants.fontSize;
        }
    }
}
